

---

Subject: Re: The IDL way: Find last non-zero value  
Posted by [penteado](#) on Mon, 24 Aug 2009 02:10:54 GMT  
[View Forum Message](#) <> [Reply to Message](#)

---

On Aug 23, 9:32 pm, Chris <beaum...@ifa.hawaii.edu> wrote:

```
> sz = size(array)
> ncol = sz[1]
> nrow = sz[2]
> nonzero = where(array ne 0)
> ind = array_indices(array, nonzero)
> sorted = sort(ind[0, *])
> result = fltarr(nrow) - 1
> result[ind[1, sorted]] = ind[0, sorted]
```

That gave me this idea:

```
function lastnz2d,array
;Returns a vector with the index of the last nonzero element
;of each line of array (-1 if that line only has zeroes)
sz=size(array)
;Find all nonzero elements in array
nonzero=where(array ne 0,count)
;Get out if all elements of array are zero
if (count eq 0) then return,replicate(-1,sz[2])
;Find the row/column number of each nonzero element
ind=array_indices(array,nonzero)
;Initialize the result vector
res=lonarr(sz[2])-1L
;The repeated numbers in the second column of ind are
;the elements in the same line, so find the last occurrence
;of each sequence of repeated numbers
uni=uniq(ind[1,*])
;The result of uniq is the row number in ind of the last nonzero
element
;of each row in array
res[ind[1,uni]]=ind[0,uni]
return,res
end
```

It is the same up to the use array\_indices to get row and column numbers into ind. For each row of array with nonzero elements, the corresponding rows of ind will have the same number in the second column (since it is the row number in array), with the increasing column numbers in the first column. So I use uniq on the second column of ind to retrieve every row in ind that corresponds to a last nonzero element in a row of array.

---